Dr. Straight's Maple Examples
Example II: Calculus Operations and Functions
Functions Covered: diff, dsolve, implicitdiff, implicitplot, int, limit, odeplot, piecewise, plot, solve, subs, sum, taylor

Let's begin by evaluating several limits.

> $limit\left(\frac{\mathrm{sqrt}(x^2+4)-2}{x^2}, x=0\right);$

$$\frac{1}{4} \quad (1)$$

> $limit\left(\frac{1-\cos(t)}{t}, t=0\right);$

$$0 \quad (2)$$

> $limit\left(\left(1+\frac{1}{n}\right)^n, n=\infty\right);$

$$e \quad (3)$$

> $limit\left(\frac{6x^3+5}{2x^3+7x^2+9x+11}, x=\infty\right);$

$$3 \quad (4)$$

Let's investigate the vertical and horizontal asymptotes of $f(x) = \mathrm{sqrt}(4x^2 + 3)/(2x - 1)$.

> $f := x \rightarrow \frac{\mathrm{sqrt}(4 \cdot x^2+3)}{2 \cdot x-1};$

$$f := x \rightarrow \frac{\sqrt{4x^2+3}}{2x-1} \quad (5)$$

> $limit\left(f(x), x=\frac{1}{2}, left\right);$

$$-\infty \quad (6)$$

> $limit\left(f(x), x=\frac{1}{2}, right\right);$

$$\infty \quad (7)$$

> $limit(f(x), x=-\infty);$

$$-1 \quad (8)$$

> $limit(f(x), x=\infty);$

$$1 \quad (9)$$

Thus, the line $x = 1/2$ is a vertical asymptote and the lines $y = -1$ and $y = 1$ are horizontal asymptotes.

Next, we use the **piecewise** command to define a "piecewise" function and investigate its continuity at $x = 1$.

> $g := x \rightarrow piecewise(x < 1, 3 \cdot x+1, x \geq 1, 5 - x^2);$

$$g := x \rightarrow piecewise(x < 1, 3x+1, 1 \leq x, 5-x^2) \quad (10)$$

> $g(1);$

$$4 \quad (11)$$

> $limit(g(x), x=1, left);$

$$4 \quad (12)$$

> $limit(g(x), x = 1, right);$

$$4 \tag{13}$$

Therefore, $g(x)$ is continuous at $x = 1$.

A more detailed discussion of the **plot** command is coming up later; for now, I want to plot $y = g(x)$ to make sure the piecewise command is working as expected.

> $plot(g(x), x = -3..5);$



The **diff** function is used for differentiation. Here are a couple of simple examples:

> $diff(5 \cdot x^3 + 4 \cdot x^2 + 7, x);$

$$15\,x^2 + 8\,x \tag{14}$$

> $diff(\tan(x), x);$

$$1 + \tan(x)^2 \tag{15}$$

> $simplify(\%);$

$$\frac{1}{\cos(x)^2} \tag{16}$$

Apparently, Maple has something against the secant function. Also, be careful! When Maple says

$$\tan(x)^2$$

it means $(\tan x)^2$, not $\tan(x^2)$.

One can also use the differentiation template found on the "Expression" palette:

> $\frac{d}{dx}(\tan(x))^2;$

$$2\tan(x)\left(1+\tan(x)^2\right) \quad (17)$$

I don't like Maple's syntax for higher-order derivatives; I prefer to iterate *diff*:

> $diff(diff((\sin(x))^2, x), x);$

$$2\cos(x)^2 - 2\sin(x)^2 \quad (18)$$

or

> $\frac{d^2}{dx^2}(\sin(x));$

$$-\sin(x) \quad (19)$$

The *diff* function is also used for partial differentiation:

> $f := (x, y) \rightarrow x^3 + x^2 \cdot y + e^y;$

$$f := (x, y) \rightarrow x^3 + x^2\, y + e^y \quad (20)$$

> $diff(f(x, y), x);$

$$3x^2 + 2xy \quad (21)$$

> $diff(f(x, y), y);$

$$x^2 + e^y \quad (22)$$

> $diff(diff(f(x, y), y), x);$

$$2x \quad (23)$$

or

> $\frac{\partial}{\partial x}\left(\frac{\partial}{\partial y} f(x, y)\right);$

$$2x \quad (24)$$

or

> $\frac{\partial^2}{\partial x\, \partial y} f(x, y);$

$$2x \quad (25)$$

Look above at the graph of $y = g(x)$. It appears that $g(x)$ is not differentiable at $x = 1$. Let's see what Maple says.

> $g(x);$

$$\begin{cases} 3x+1 & x<1 \\ 5-x^2 & 1 \le x \end{cases} \quad (26)$$

> $diff(g(x), x);$

$$\begin{cases} 3 & x<1 \\ \mathit{undefined} & x=1 \\ -2x & 1<x \end{cases} \quad (27)$$

Cool!

To perform implicit differentiation, use the **implicitdiff** function. For example, suppose we want to find the slope of the line tangent to the ellipse $x^2 + 4y^2 = 25$ at the point (3,-2). We first use *implicitdiff* to get dy/dx:

> $implicitdiff(x^2+4\cdot y^2=25, y, x);$

$$-\frac{1}{4}\frac{x}{y} \tag{28}$$

We then substitute $x = 3$ and $y = -2$ into this expression to get the slope m of the tangent line:

> $subs([x=3, y=-2], \%);$

$$\frac{3}{8} \tag{29}$$

So, the equation of the tangent line is $3x - 8y = 25$. Again, a more detailed discussion of graphing is a "coming attraction," but let's use the **implicitplot** command to graph the ellipse and the tangent line.

> $with(plots) : implicitplot([x^2+4\cdot y^2=25, 3\cdot x-8\cdot y=25], x=-3..9, y=-4..2);$



Note that this command is in the "plots" package, so we need to open the package before invoking the command.

Let's move on to integration. To integrate something, we use the **int** function. There are two forms, one for indefinite integration, the other for definite integration. Here are some examples:

> $int\left(\frac{1}{\sqrt{a^2+x^2}}, x\right);$

$$\ln\left(x+\sqrt{a^2+x^2}\right) \tag{30}$$

Note that Maple doesn't bother with the "plus a constant."

> $int\left(\frac{1}{\sqrt{9+x^2}}, x=0..4\right);$

$$\frac{1}{2}\,\frac{-(-6\ln(2)+2\ln(3))\sqrt{\pi}+2\sqrt{\pi}\ln\left(\frac{9}{8}\right)}{\sqrt{\pi}} \qquad \textbf{(31)}$$

Why did Maple get such a weird answer? I don't know, but let's try simplifying it:

> $simplify(\%);$

$$\ln(3) \qquad \textbf{(32)}$$

That's better! We can also integrate using the templates provided on the "Expression" palette:

> $\int \frac{1}{a^2+x^2}\,\mathrm{d}x;$

$$\frac{\arctan\left(\frac{x}{a}\right)}{a} \qquad \textbf{(33)}$$

> $\int_0^{\sqrt{3}} \frac{1}{1+x^2}\,\mathrm{d}x;$

$$\frac{1}{3}\,\pi \qquad \textbf{(34)}$$

Maple can even handle improper integrals:

> $int\left(\mathrm{e}^{-x^2}, x=-\infty..\infty\right);$

$$\sqrt{\pi} \qquad \textbf{(35)}$$

> $\int_0^1 \frac{1}{\sqrt{x}}\,\mathrm{d}x;$

$$2 \qquad \textbf{(36)}$$

Maple's good, but it can't do everything. For example:

> $int\left(\sqrt{1+\ln(x)}, x=1..2\right);$

$$-1-\frac{1}{2}\,\mathrm{I}\,\mathrm{e}^{-1}\sqrt{\pi}\,\mathrm{erf}(\mathrm{I})+2\sqrt{\ln(2)+1}+\frac{1}{2}\,\mathrm{I}\,\mathrm{e}^{-1}\sqrt{\pi}\,\mathrm{erf}\left(\mathrm{I}\sqrt{\ln(2)+1}\right) \qquad \textbf{(37)}$$

"erf" is the "error function." If Maple is unable to evaluate a definite integral *exactly*, ask yourself if you would be happy with an approximation. If so, you can apply *evalf* to your *int*:

> $evalf\left(int\left(\sqrt{1+\ln(x)}, x=1..2\right)\right);$

$$1.174326172 \qquad \textbf{(38)}$$

or give the limits of integration as type "float:"

> $int\left(\sqrt{1+\ln(x)}, x=1.0..2.0\right);$

$$1.174326172 \qquad \textbf{(39)}$$

Either way, Maple will perform "numeric" integration to find an approximation.

I don't like Maple's syntax for multiple integrals. Instead, I recommend using the integral template to get what you want.

> $\int_0^2 \int_0^3 (25 - x^2 - 4 \cdot y^2)\, dx\, dy;$

$$100 \tag{40}$$

> $\int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \int_0^{\cos(2\theta)} r\, dr\, d\theta;$

$$\frac{1}{8}\pi \tag{41}$$

In an earlier example, we saw the *sum* function in action. In calculus, we consider infinite sums, or "infinite series." The **sum** function can be used to determine whether a series converges or not. Here are some well-known convergent series:

> $sum\left(\left(\frac{1}{2}\right)^n, n = 0 .. \infty\right);$

$$2 \tag{42}$$

> $sum\left(\frac{1}{n^2}, n = 1 .. \infty\right);$

$$\frac{1}{6}\pi^2 \tag{43}$$

> $sum\left(\frac{(-1)^{n+1}}{n}, n = 1 .. \infty\right);$

$$\ln(2) \tag{44}$$

What happens if we enter a divergent series, such as the harmonic series?

> $sum\left(\frac{1}{n}, n = 1 .. \infty\right);$

$$\infty \tag{45}$$

Maple says: divergent!

The **taylor** function is used for Taylor series. Specifically,
taylor(f(x),x=a,n)
returns the Taylor polynomial of degree n - 1 for $f(x)$ about $x = a$. Here are some examples:

> $taylor(e^x, x = 0, 5);$

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + O(x^5) \tag{46}$$

> $taylor(\ln(x), x = 1, 6);$

$$x - 1 - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + \frac{1}{5}(x-1)^5 + O((x-1)^6) \tag{47}$$

> $taylor(\sin(x), x = 0, 11);$

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 + O(x^{11}) \tag{48}$$

> $h := x \to x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9;$

$$h := x \to x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 \tag{49}$$

> $h\left(\frac{\pi}{6}\right);$

$$\frac{1}{6}\pi - \frac{1}{1296}\pi^3 + \frac{1}{933120}\pi^5 - \frac{1}{1410877440}\pi^7 + \frac{1}{3656994324480}\pi^9 \quad \textbf{(50)}$$

> $evalf(\%);$

$$0.5000000003 \quad \textbf{(51)}$$

> $evalf\left(\sin\left(\frac{\pi}{6}\right)\right);$

$$0.5000000000 \quad \textbf{(52)}$$

Powerful stuff!

To solve differential equations, we employ the **dsolve** function. This is not my area of expertise, so I'll be brief, and encourage others to add their "two cents."

For our first example, we solve the first-order ODE $dy/dt = y - t^2 + 1$ subject to the initial condition $y(0) = 1/2$:

> $dsolve\left(\left\{diff(y(t), t) = y(t) - t^2 + 1, y(0) = \frac{1}{2}\right\}\right);$

$$y(t) = 1 + 2t + t^2 - \frac{1}{2}e^t \quad \textbf{(53)}$$

For this initial-value problem, Maple was able to obtain an exact solution. But, if not, we could ask Maple to solve the ODE numerically, storing the solution in *soln*:

> $soln := dsolve\left(\left\{diff(y(t), t) = y(t) - t^2 + 1, y(0) = \frac{1}{2}\right\}, numeric\right);$

$$soln := \textbf{proc}(x\_rkf45) \; ... \; \textbf{end proc} \quad \textbf{(54)}$$

We can then plot the solution using the **odeplot** command:

> $with(plots) : odeplot(soln, [t, y(t)], t = 0..2);$



Since that worked out, let's try solving the second-order equation $y'' + y' = t + 2$, subject to the initial conditions $y(0) = 1$ and $y'(0) = 0$:

> $diffeq := \frac{d^2}{d\,t^2}\, y(t) + \frac{d}{d\,t}\, y(t) = t + 2;$

$$diffeq := \frac{d^2}{dt^2}\, y(t) + \frac{d}{dt}\, y(t) = t + 2 \quad \textbf{(55)}$$

> $dsolve(diffeq);$

$$y(t) = \frac{1}{2}\, t^2 - e^{-t}\, \_C1 + t + \_C2 \quad \textbf{(56)}$$

Note that we get the general solution: $y = t^2/2 - ae^{-t} + t + b$. To get a particular solution, we need some initial conditions.

> $initcond := y(0) = 1, D(y)(0) = 0;$

$$initcond := y(0) = 1, D(y)(0) = 0 \quad \textbf{(57)}$$

> $dsolve(\{diffeq, initcond\});$

$$y(t) = \frac{1}{2}\, t^2 + e^{-t} + t \quad \textbf{(58)}$$

>